

情報処理 2 第 3 回

十進 BASIC (3) グラフィックス入門

かつらだ まさし
桂田 祐史

2007 年 5 月 1 日

この授業用の WWW ページは <http://www.math.meiji.ac.jp/~mk/syori2-2007/>

数学のためにコンピューターに計算をさせる場合、数値計算 (Numerical computation), 数式処理 (Symbolic computation), グラフィックス (Graphics, 可視化 visualization) が 3 本柱と言われます。今日は十進 BASIC で簡単なグラフィックスをやってみましょう。

1 連絡事項

- 十進 BASIC のオンライン・ヘルプですが、近日中に見られるようになる見通しです。もうしばらく待って下さい。
- 5 月 15 日 (火曜)3 限に数学科学生を対象に『インターネット講習会』を行います¹。
 - － 既にインターネット講習会を受講済みの人は出席する必要はありません (出席扱いになります)。
 - － アンケートで講習会を受講済みと答えただけ、その後インターネット講習会を受講した人は、5 月 11 日 (金曜) までに桂田 (mk@math.meiji.ac.jp) 宛に「連絡：インターネット講習会」という表題でメールをして、本文に学年・組・番号・氏名と、「すでにインターネット講習会を受講済みですので 5 月 15 日は出席しません。」と連絡して下さい。
 - － 数学科の学生で、この講義を受講していない人も、このインターネット講習会を受講しても構いません。その場合、準備の都合上 (書類の枚数を知りたい)、mk@math.meiji.ac.jp 宛にメールで連絡をもらえると助かります。友達に教えてあげてください。

当日は

遅刻厳禁

です。講習会に少しでも遅れた場合は受講を認めませんのでよくよく注意してください。

¹先日のアンケートで 20 人以上の人が未受講であることが分かったので、是非とも実施、と思っています。

2 十進 BASIC でグラフィックスを

2.1 イントロ

汎用のプログラミング言語で、その規格の中にグラフィックスを含んでいるものは実は珍しく (Java と BASIC くらいしか思い浮かばない...)、十進 BASIC はグラフィックスをするのはオススメです。

まずチュートリアル (印刷配布してある文書) (<http://hp.vector.co.jp/authors/VA008683/tutorial/contents.htm>) の 2.6 節を読んでみましょう。

2.2 学外にアクセスできない人向け

既に配った十進 BASIC の説明書も忘れた (持って来て下さい)、インターネット講習会を受講していないので <http://hp.vector.co.jp/authors/VA008683/tutorial/contents.htm> も見られないという人のために (本当は探せば `turotrial.pdf` あるはずですが)、「2.6 グラフィックス」の例題プログラムをここに引用しておきます。

例 14 (関数のグラフ)

```
10 DEF f(x)=x^3-3*x+1
20 SET WINDOW -4,4,-4,4
30 DRAW GRID
40 FOR x=-4 TO 4 STEP 0.1
50   PLOT LINES: x,f(x);
60 NEXT x
70 END
```

(要点: 1 変数実数値関数のグラフ $y = f(x)$ の描き方の定跡みたいなプログラム)

例 15 (OPTION ANGLE DEGREES, DRAW GRID())

```
10 OPTION ANGLE DEGREES
20 DEF f(x)=sin(x)
30 SET WINDOW -360,360,-4,4
40 DRAW GRID(90,1)
50 FOR x=-360 TO 360
60   PLOT LINES: x,f(x);
70 NEXT x
80 END
```

(要点: 三角関数の単位を度にする OPTION ANGLE DEGREES と DRAW GRID(.) の使い方)

例 16 (“PLOT LINES” で線を切る)

2つの関数のグラフを描くために、途中で「切る」必要がある。

```
100 DEF f(x)=x^2
110 DEF g(x)=x^3
120 SET WINDOW -4,4,-4,4
130 DRAW GRID
140 FOR x=-4 TO 4 STEP 0.1
150   PLOT LINES: x,f(x);
160 NEXT x
170 PLOT LINES
180 FOR x=-4 TO 4 STEP 0.1
190   PLOT LINES: x,g(x);
200 NEXT x
210 END
```

(要点: “PLOT LINES: x,y;” は線をつなげていくので、セミコロンなしの “PLOT LINES” で線を切る。)

例 17 (パラメーター曲線)

```
10 OPTION ANGLE DEGREES
20 DEF f(t)=3*COS(t)
30 DEF g(t)=2*SIN(t)
40 SET WINDOW -4,4,-4,4
50 DRAW grid
60 FOR t=0 TO 360
70   PLOT LINES: f(t),g(t);
80 NEXT t
90 END
```

(要点: パラメーター曲線 $x = f(t)$, $y = g(t)$ の描き方の定跡みたいなプログラム)

例 18 (極方程式表示の曲線)

```
10 DEF f(t)=SIN(2*t)
20 SET WINDOW -1,1,-1,1
30 DRAW grid
40 FOR t=0 TO 2*PI STEP PI/360
50 PLOT LINES: f(t)*COS(t), f(t)*SIN(t);
60 NEXT t
70 END
```

(要点: 極方程式 $r = f(\theta)$ は、 $x = f(\theta) \cos \theta$, $y = f(\theta) \sin \theta$ として単なるパラメーター曲線)

例 19 (直交座標から極座標への変換)

```
100 SET WINDOW -4,4,-4,4
110 DRAW grid
120 FOR t=0 TO 2*pi STEP pi/180
130   LET x=cos(t)+1
140   LET y=sin(t)
150   LET r=x^2+y^2
160   LET a=ANGLE(x,y)*2
170   PLOT LINES: r*cos(a),r*sin(a);
180 NEXT t
190 END
```

($x = r \cos t$, $y = r \sin t$ の x, y から r, t を求めるには、 $r = \text{SQRT}(x^2+y^2)$, $t = \text{ANGLE}(x,y)$ とすれば OK)

余談 2.1 C 言語のプログラムならば、 $r = \text{sqrt}(x*x+y*y)$; (または $r = \text{hypot}(x,y)$;) そして

$t = \text{atan2}(y, x)$; とするところ。 $\text{atan2}()$ という得体の知れない名前の関数よりは $\text{ANGLE}()$ の方が覚えやすくよい、と私は思います。 ■

2.3 まとめ

- `SET WINDOW x_1, x_2, y_1, y_2` は座標系の設定
- `DRAW GRID` はグリッドを描く
`DRAW GRID(x, y)` というように幅も指定できる
- `PLOT LINES: x, y` ; 次に実行される `PLOT LINES` 命令で指定される点との間を線分で結ぶ。線を区切るには、単に `PLOT LINES` とする。
- 1 変数関数のグラフ、パラメーター曲線は描くのが簡単 (例題プログラム参照)。
- 点を描く `PLOT POINT: x, y` 命令, 文字を描く `PLOT TEXT , AT x, y : "文字列"` 命令, 多角形領域を塗り潰す `PLOT AREA: $x_1, y_1; \dots; x_n, y_n$` 命令などもある。これらの命令の使い方についてはオンライン・ヘルプを参照。
- 色も指定できる。
`SET POINT COLOR c` , `SET LINE COLOR c` , `SET AREA COLOR c` , `SET TEXT COLOR c` 等々。ここで c は色番号で、以下のように定義されています。
0 白, 1 黒, 2 青, 3 緑, 4 赤, 5 水色, 6 黄色, 7 赤紫, 8 灰色, 9 濃い青, 10 濃い緑, 11 青緑, 12 えび茶, 13 オリーブ色, 14 濃い紫, 15 銀色, ...
- グラフのウィンドウの「ファイル」メニューで、「名前をつけて保存」から描いたグラフィックスをファイルに保存できますが、「ファイルの種類 (T)」で、BitMap でなく、JPEG や GIF フォーマットを選ぶことを勧めます (画像ファイルはサイズが大きくなりがちで、特別の理由がない限り、何らかの方法で圧縮したフォーマットを選ぶべきです)。

3 課題 3A

十進 BASIC で円を描く²プログラムと描いた図を送って下さい (図の形式は JPEG にして下さい)。

宛先は syori2@math.meiji.ac.jp, Subject (表題) は「情報処理 2 課題 3A」で、他の注意はもう繰り返しません。


これは出席代りで今日 (5 月 1 日) 中に送ること。

²昔、パソコンで BASIC 華やかなりし頃は「(楕)円を描く命令」が備わっていました。でも十進 BASIC にはありません。ないけれども“簡単に描ける”はず。


4 課題 3B

次のうちのいずれかを行うプログラムを作成して下さい。プログラムとその説明、描いた図そのものも (ファイルのフォーマットは JPEG) 送って下さい。締め切りは (連休があるので) 5月14日 (月) とします。Subject (表題) は「情報処理2 課題 3B」です。

- (1) 銀色の星を描く (図の星形の中を銀色に塗る)。



— 大変だったら塗り潰しできなくても構いません。また右図のように五角形を塗るのもよいです。PLOT AREA 以外に PAINT や MAT PLOT AREA などが利用できます。

- (2) 自己相似図形として有名な Koch 曲線 (の近似) を描く。 有名なので探せばそのものズバリのプログラムも見つかるでしょうが、自分で考えて見ませんか? サブルーチンの再帰呼び出しを使えば割と簡単です。

A 課題 3B への一つのヒント: タートルグラフィックス

一時期、プログラミングの入門用の言語として、LOGO³ というものが注目されたことがあります。その特徴であるタートル・グラフィックスを利用すれば、まだ座標系の概念を学んでいないような小学生にも簡単に図形を描くプログラムが作れるということで「ちょっと面白い」話でした。

タートル・グラフィックスとは、キャンパスの中を亀 (タートル) が歩く軌跡として図形を描画するというものです。亀に伝えられる指令として、「進む (指定された歩数だけ)」、「戻る (指定された歩数だけ)」、「右に曲がる (指定された角度だけ)」、「左に曲がる (指定された角度だけ)」、「ペンを下ろす」、「ペンをあげる」などがあります。

例えば、「50 歩進み、90° 右に曲がる」を 4 回繰り返すことで 1 辺の長さが 50 歩の正方形を描くことができます。

³<http://ja.wikipedia.org/wiki/LOGO>

```

REM TURTLE.BAS --- 亀よ正方形と正三角形を描いておくれ
OPTION ANGLE DEGREES
SUB right(t)
    LET direction=direction-t
END SUB
SUB left(t)
    LET direction=direction+t
END SUB
SUB walk(s)
    PLOT LINES: xp,yp;
    LET xp=xp+s*COS(direction)
    LET yp=yp+s*SIN(direction)
    PLOT LINES: xp,yp
END SUB
SUB jump(s)
    LET xp=xp+s*COS(direction)
    LET yp=yp+s*COS(direction)
    PLOT LINES: xp,yp
END SUB
REM ----- start -----
LET L=100
SET WINDOW -1.1*L,1.1*L,-1.1*L,1.1*L
REM 亀の現在位置 (xp,yp), 方向 direction (x軸となす角)
LET direction=0
LET xp=0
LET yp=0
FOR i=1 TO 4
    CALL walk(L)
    CALL right(90)
NEXT i
CALL right(10)
SET LINE COLOR "red"
FOR i=1 TO 3
    CALL walk(L/2)
    CALL right(120)
NEXT i
END

```

